<commit_before>############################################################################
# CMakeLists.txt file for building ROOT bindings/pyroot package
############################################################################

ROOT_ADD_CXX_FLAG(CMAKE_CXX_FLAGS -fno-strict-aliasing)
ROOT_ADD_CXX_FLAG(CMAKE_CXX_FLAGS -Wno-parentheses-equality)
ROOT_ADD_CXX_FLAG(CMAKE_CXX_FLAGS -Wno-cast-function-type)
ROOT_ADD_CXX_FLAG(CMAKE_CXX_FLAGS -Wno-deprecated-register)

if(WIN32)
  set(ROOTULIBS -include:_G__cpp_setupG__Net
                -include:_G__cpp_setupG__IO
                -include:_G__cpp_setupG__Tree
                -include:_G__cpp_setupG__Thread
                -include:_G__cpp_setupG__MathCore)
endif()

set(PYROOT_HEADERS inc/TPyFitFunction.h    inc/TPython.h
                   inc/TPyArg.h            inc/TPyROOTApplication.h
                   inc/TTreeAsFlatMatrix.h inc/TPyDispatcher.h
                   inc/TPyReturn.h         inc/TPyException.h
                   inc/TPySelector.h)

set(PYROOT_PKG_DEPENDENCIES Core MathCore Net Rint Tree ROOTVecOps ROOTDataFrame)

if( CMAKE_SIZEOF_VOID_P EQUAL 4 ) # With this we exclude ROOTDataFrame on 32 bits builds
  list(REMOVE_ITEM PYROOT_HEADERS inc/TTreeAsFlatMatrix.h)
  list(REMOVE_ITEM PYROOT_PKG_DEPENDENCIES ROOTDataFrame)
endif()

ROOT_STANDARD_LIBRARY_PACKAGE(PyROOT
                              NO_INSTALL_HEADERS
                              HEADERS ${PYROOT_HEADERS}
                              DICTIONARY_OPTIONS "-writeEmptyRootPCM"
                              LIBRARIES Core MathCore Net Rint Tree
                              DEPENDENCIES ${PYROOT_PKG_DEPENDENCIES})
ROOT_LINKER_LIBRARY(JupyROOT ../JupyROOT/src/*.cxx DEPENDENCIES Core CMAKENOEXPORT)
<commit_msg>Set PyROOT compile flags without changing CMAKE_CXX_FLAGS<commit_after>############################################################################
# CMakeLists.txt file for building ROOT bindings/pyroot package
############################################################################

if(WIN32)
  set(ROOTULIBS -include:_G__cpp_setupG__Net
                -include:_G__cpp_setupG__IO
                -include:_G__cpp_setupG__Tree
                -include:_G__cpp_setupG__Thread
                -include:_G__cpp_setupG__MathCore)
endif()

set(PYROOT_HEADERS inc/TPyFitFunction.h    inc/TPython.h
                   inc/TPyArg.h            inc/TPyROOTApplication.h
                   inc/TTreeAsFlatMatrix.h inc/TPyDispatcher.h
                   inc/TPyReturn.h         inc/TPyException.h
                   inc/TPySelector.h)

set(PYROOT_PKG_DEPENDENCIES Core MathCore Net Rint Tree ROOTVecOps ROOTDataFrame)

if( CMAKE_SIZEOF_VOID_P EQUAL 4 ) # With this we exclude ROOTDataFrame on 32 bits builds
  list(REMOVE_ITEM PYROOT_HEADERS inc/TTreeAsFlatMatrix.h)
  list(REMOVE_ITEM PYROOT_PKG_DEPENDENCIES ROOTDataFrame)
endif()

ROOT_STANDARD_LIBRARY_PACKAGE(PyROOT
                              NO_INSTALL_HEADERS
                              HEADERS ${PYROOT_HEADERS}
                              DICTIONARY_OPTIONS "-writeEmptyRootPCM"
                              LIBRARIES Core MathCore Net Rint Tree
                              DEPENDENCIES ${PYROOT_PKG_DEPENDENCIES})
ROOT_LINKER_LIBRARY(JupyROOT ../JupyROOT/src/*.cxx DEPENDENCIES Core CMAKENOEXPORT)

ROOT_ADD_CXX_FLAG(_PyROOT_FLAGS -fno-strict-aliasing)
ROOT_ADD_CXX_FLAG(_PyROOT_FLAGS -Wno-parentheses-equality)
ROOT_ADD_CXX_FLAG(_PyROOT_FLAGS -Wno-cast-function-type)
ROOT_ADD_CXX_FLAG(_PyROOT_FLAGS -Wno-deprecated-register)

separate_arguments(_PyROOT_FLAGS)

target_compile_options(PyROOT PRIVATE ${_PyROOT_FLAGS})
{'fraction_non_alphanumeric': 0.1380730167444414, 'fraction_numerical': 0.002744990392533626, 'mean_word_length': 2.9437229437229435, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3804750', 'n_tokens_mistral': 1200, 'n_tokens_neox': 1120, 'n_words': 172}